
ABSTRACT 
Documents in Word format are popular in the pharmaceutical industry.  However, SAS does not provide a procedure 
to import data directly from Microsoft Word into SAS as PROC IMPORT reads data from Microsoft Excel.  This 
deficiency stimulates SAS programmers to explore the techniques of how to programmatically convert Word 
documents into SAS data.   
 
This paper will review several published solutions with pros and cons and present a creative way programmatically to 
import any Word-readable documents, such as .doc, .rtf, .txt, .xml, .html, .wpd, .wps, and .dot files, into SAS.  In 
addition, a SAS macro, %word2sas, used for the converting process will be introduced.  With this convenient and 
reliable solution, SAS programmers will become more effective and innovative to leverage MS Word and SAS for their 
daily work. 

INTRODUCTION 
Microsoft Word is standard word processing software in the pharmaceutical industry widely used to write formal 
documentations, including clinical study protocol, statistical analysis plan, analysis dataset specifications, and clinical 
study report.  Tables and listings generated from SAS in Word format can be directly inserted or included into these 
formal Word documents with the similar font type and size without manual editing.  Often times, it is desired for SAS 
programmers directly to convert data or text from the Word documents back into SAS.  For instances, it is valuable to 
replace human visual check by programmatically identifying differences between versions of Word tables/listings 
caused by database update; it is more accurate and efficient to populate derived variables with their attributes 
programmatically obtained from the Word document of the analysis dataset specifications.  However, there is no 
existing SAS procedure which will directly import Word document into SAS. 
 
To meet the challenge, SAS programmers have developed a few ways to read data from Word with SAS 
(Hagendoorn et al., 2006; Xu and Zhou, 2007; Huang and Lynn, 2008; Tran, 2008).  Each method has its own pros 
and cons. 
 
Hagendoorn et al. (2006) developed a utility macro tool that converts RTF-formatted table/listing back to SAS for the 
purpose of quality control (QC), comparing it against the data independently programmed with PROC COMPARE.  
There are many limitations with this method.  Because the tool reads the RTF file with the INFILE statement in a data 
step to import the data from the table body based on the RFT control words, intensive maintenance to the macro is 
needed to remove different special symbols, and sub-/super-scripts, and to fit different column settings and data 
presentations.  Artificially searching the specific RTF tagsets could be cumbersome and unreliable.  In addition, table 
titles, column headers, and footnotes, and special characters are not imported for QC comparison.   
 
A much easy way introduced by Xu and Zhou (2007) is to use the INFILE statement to convert the plain text file 
programmatically saved by Word via DDE (Dynamic Data Exchange) into SAS.  This method can import any Word-
readable document into SAS without concerning RTF tagsets.  However, it can only be valid in the platform with Word 
installed so that the DDE linkage can be established between Word and SAS.  A manual click is necessary to close a 
pop-up window, if a Word document contains any special symbol characters, including sub-/super-scripts, during the 
step of saving as a plain text file, because there is no option available in the WordBasic FileSaveAs command to 
disable the pop-up window.  
 
A similar approach of importing a text file with the INFILE statement into SAS was used by Huang and Lynn (2008) for 
obtaining titles and footnotes from mockup tables to eliminate the step of retyping them in table production.  However, 
the converting process requires a manual step to edit and save the Word mockup table into a plain text file prior to 
calling the macro.  This manual process could introduce potential human error and be time-consuming when dealing 
with hundreds of tables.  
 
Tran (2008) presented a method using Perl Regular Expression to express the RTF Regular Expression in order to 
strip the RTF tagsets.  This Regular Expression approach provides more reliable outcome than the method provided 
by Hagendoorn et al. (2006).  However, it requires SAS v9.1.3 or above which provides a set of PRX CALL routines 
and functions to handle Perl Regular Expression.  Potential maintenance to the RTF Regular Expression may be 

needed due to a new Word release or RTF version change.  In addition, Tran’s %RTFparser utility tool can be easily 
improved to parser any Word-readable documents by adding a step with the DDE linkage as employed by Xu and 
Zhou (2007) to save non-RTF files in RTF prior to pursuing the Regular Expression. 
 
This paper introduces another useful alternative to import data from any Word-readable documents into SAS.  The 
entire process involves SAS, Word, and Excel which is automated by a SAS macro called %word2sas.  It will 
overcome any drawback of converting a Word file into plain text.  More importantly, it will import the entire contents 
with reserving the document layout and structures, including tables with multiple columns, into SAS data. 

TECHNIQUE & MECHANISM 
The common manual practice to read a particular Word table into SAS is to copy the target content from Word and 
paste it in Excel, then save it as .txt (tab delimited) or .xls (Excel) file which can be easily imported into SAS with the 
INFILE statement or PROC IMPORT.  The entire process is divided into 4 important steps: 
 

1. Open the target file in Word and copy the content 
2. Open Excel and paste the content in Excel 
3. Save it as delimited or Excel .xls file 
4. Read the content into SAS with INFILE statement or PROC IMPORT 

 
To automate these steps, the DDE solution is leveraged to build the communication bridge between SAS and Word or 
Excel.  WordBasic command (Microsoft Corporation, 1999) and Excel v4.0 macro function (Microsoft Corporation, 
2000) can be sent from SAS via DDE to enable SAS to take control of Word and Excel. 

OPEN THE TARGET FILE IN WORD AND COPY THE CONTENT 
In order for a client/server communication link to be established, both SAS and Word must be running.  Therefore, for 
the first iteration, it is necessary to programmatically launch Word from a SAS session.  There are several techniques 
available to launch Word from SAS.  The simplest one is the following statement: 
 
   %let rc=%sysfunc(system(start winword)); 
 
Prior to executing this statement, there are two system options needed: NOXWAIT and NOXSYNC.  The NOXWAIT 
option specifies that the DOS command prompt window disappears without your having to type EXIT when the 
process is finished, while the NOXSYNC specifies that the process should execute asynchronously.  That is, control is 
returned immediately to the SAS System and the command continues executing without interfering with your SAS 
session. 
 
To communicate with Word from SAS, the following FILENAME statement is used to establish the linkage between 
SAS and Word via the DDE triplet: 
 
   filename word DDE 'Winword|System'; 
 
The next step is to open the target Example.doc file by sending the FileOpen WordBasic command to Word with a 
data _null_ step.  Once the file is open in Word, select and copy the entire document by sending EditSelectAll 
and EditCopy shown in the following: 
 
   data _null_; 
      file word; 
      put '[FileOpen .Name = "' "C:\PharmaSUG2009\Example.doc" '"]'; 
      put "[EditSelectAll]"; 
      put "[EditCopy]"; 
      put '[FileClose]'; 
   run; 
     
It is necessary to programmatically close the document with the FileClose command to clean the space to avoid 
multiple Word sessions open when converting multiple Word files.   

OPEN EXCEL AND PASTE THE CONTENT IN EXCEL 
The same approach as invoking Word is used to start a new Excel session: 
      
   %let rc=%sysfunc(system(start excel)); 

To communicate with Excel from SAS, the following FILENAME statement is used to establish the linkage between 
SAS and Excel via the DDE triplet: 
  
   filename excel dde 'Excel|System'; 
 
Once Excel is running, the next step is to paste the content copied from Word to Excel.  This process in Excel can be 
easily automated from SAS with the following code: 
 
   data _null_; 
      file excel; 
      put '[Error(false)]'; 
      put '[Paste]'; 
   run; 
 
The Excel Error(false) macro function clears all error-checking and keeps Excel from displaying pop-up window. 
Other commands can be continuously sent and executed even if an error is encountered. 

SAVE AS DELIMITED OR EXCEL .XLS FILE 
To save the pasted content in either delimited or Excel format is depending on which method will be used to import it 
into SAS.  The following code saves it as the Example.xls file under the C:\PharmaSUG2009 directory: 
 
   data _null_; 
      put '[Error(false)]'; 
      put '[SAVE.AS("C:\PharmaSUG2009\Example.xls")]';  
      put '[Select.Last.Cell()]';  
      put '[Copy]';  
      put '[Quit]';  
   run; 
 

 equivalent to choosing the Save As command from the File menu which offers many choices of different file types 
(Microsoft Press, 1992).  By default, the example.xls is saved as Excel .xls file.   equivalent to choosing the Special 
button from the Goto dialog box and selecting the Last Cell Option.   equivalent to choosing the Copy command 
from the Edit menu.   equivalent to choosing the Exit command from the File menu in Excel for Windows.  It quits 
Excel, closes any open documents, and terminates the Copy command.  The purpose of , , and  is to clear the 
content copied from Word and stored in the Clipboard to prevent potential problem from pasting it into a wrong place 
later.  

READ THE CONTENT INTO SAS WITH INFILE STATEMENT OR PROC IMPORT 
After the text or data is saved into SAS-readable file, it is time to read it into SAS by either using the infile 
statement or proc import depending on which SAS version you are using.  For SAS v8.2 or below, the proc 
import procedure will not be able to handle those columns with mixed numeric and character values because the 
field type is determined by scanning the first 20 rows, by default.  Therefore, it is more reliable to save the file as 
common (.csv) or tab (.txt) delimited file for the convenience of using the INFILE statement or using more 
sophisticated method described by Zhou (2002).  With SAS v9.1 or above, SAS improved the PROC IMPORT 
procedure by adding the mixed option to solve the issue as shown in the following code:   
 
   proc import out=temp datafile= "C:\PharmaSUG2009\Example.xls" replace;  
        getnames=no;  
        mixed=yes; 
   run; 
 
The SAS dataset called temp is created and saved in the work library for further data manipulation or preparation in 
SAS.  At this point, the intermediate file Example.xls can be deleted: 
 
   filename fd "C:\PharmaSUG2009\Example.xls"; 
   %let rc=%sysfunc(fdelete(fd)); 
   filename fd clear; 

%WORD2SAS MACRO 
To facilitate and automate the above discussed steps from opening the Word file to reading the Excel file into SAS, a 
SAS macro called %word2sas was developed for SAS v9.1 or above (see Appendix for details).  User can easily 
extend the macro to fit other SAS versions by adding the data step with the INFILE statement.  There are only two 
keyword parameters:  
 
In: Define the path and name of the input Word file, e.g., C:\PharmaSUG2009\Example.doc. 
Out: Define the SAS dataset name.  By default, the dataset name is temp. 
 
Below is a simple macro call to %word2sas: 
 
      %word2sas(in= C:\PharmaSUG2009\Example.xls, out=Example); 

CONCLUSION 
Compared to other solutions mentioned in the Introduction section, the %word2sas solution can convert any Word-
readable files, including .doc, .rtf, .txt, .xml, .html, .wpd, .wps, and .dot files, into SAS data.  Because there is no file 
format conversion involved in the process, the integrity of keeping the original content (including special characters) 
and table structures is guaranteed without any manual pre-process.  The only drawback of this method is that the 
process is based on the DDE technology which may not be supported by Microsoft Corporation in future MS Office 
releases.  By that time, we hope that PROC IMPORT will be able to directly communicate with MS Word. 

APPENDIX 
 
%macro word2sas(in=, out=temp); 
   options noxwait noxsync; 
   %let rc=%sysfunc(system(start winword)); 
 
   filename word DDE 'Winword|System'; 
 
   data _null_; 
      file word; 
      put '[FileOpen.Name = "' "&in" '"]'; 
      put "[EditSelectAll]"; 
      put "[EditCopy]"; 
      put '[FileClose]'; 
      put '[AppMinimize]'; 
   run; 
          
   %let rc=%sysfunc(system(start excel)); 
 
   filename excel dde 'excel|system'; 
 
   data _null_; 
      file excel; 
      put '[error(false)]'; 
      put '[paste]'; 
      put '[SAVE.AS("C:\PharmaSUG2009\Example.xls")]'; 
      put '[Select.Last.Cell()]'; 
      put '[Copy]'; 
      put '[quit]'; 
   run; 
 
   proc import out=&out datafile= "C:\PharmaSUG2009\Example.xls" replace;  
      getnames=no; 
      mixed=yes; 
   run; 
 
   filename fd "C:\PharmaSUG2009\Example.xls"; 
   %let rc=%sysfunc(fdelete(fd)); 
   filename _all_ clear; 
%mend word2sas; 
 


